



MICROSOFT EXCEL

TECHNIQUES AVANCEES POUR CALCULS DYNAMIQUES

Mise en pratique – Apprendre au travers d'exercices

18 pages



**Livre
Gratuit**

**Support de formation
à télécharger**



Techniques et calculs avancés



TECHNIQUES AVANCEES DE CALCUL	2
IMBRICATIONS DE FONCTIONS SI	2
ET, OU : COMMENT RECOUPER LES CONDITIONS	3
LA CONCATENATION	6
CONCATENATION PAR FONCTION DE CALCUL	7
RESULTAT STATISTIQUE ET CONCATENATION	8
LA RECHERCHE D'INFORMATIONS	10
REFERENCES RELATIVES - REFERENCES ABSOLUES	12
LES CALCULS CONDITIONNELS	15
CONDITIONS ET REFERENCES ABSOLUES	17



Techniques avancées de calcul

Dans ce deuxième petit ouvrage, nous poursuivons les travaux d'apprentissage réalisés au travers du premier livre. Il s'agit d'aller plus loin et d'exploiter la puissance d'**Excel**. Nous abordons des techniques plus complexes. Des objectifs sont à observer à l'issue. Nous devons bâtir des **formules** entièrement **dynamiques**. Nous ne devons jamais refaire deux fois le même **calcul**. Pour une meilleure compréhension, nous proposons de poursuivre cet apprentissage par la pratique.

- Télécharger le [classeur techniques-avancees-calculs.xlsx](#) en cliquant sur son lien,
- L'ouvrir dans **Excel** et activer la **feuille Conditions**,

Nous récupérons le classeur des exercices précédents avec les travaux, là où nous les avons laissés.

Imbrications de fonctions si

Seule, la **fonction Si** ne permet de vérifier qu'une seule condition. De fait, elle ne peut envisager que deux possibilités. Certains problèmes professionnels sont souvent bien plus complexes. Les cas à considérer sont plus nombreux. C'est l'**imbrication des fonctions conditionnelles** (Sinon Si) qui permet d'énumérer les critères en cascade. A chaque nouveau critère ainsi posé, deux nouvelles possibilités s'ouvrent. La syntaxe de l'imbrication est la suivante :

`=Si(Critère1 ; Alors_Action1 ; Si(Critère2 ; Alors_Action2 ;... ; Si(CritèreN ; Alors_ActionN ; Sinon_ActionN)...))`

L'imbrication n'a pas de limite si ce n'est la lisibilité et la compréhension qui se dégradent au fur et à mesure que la formule grandit. Lorsque le nombre de critères devient trop important, il faut envisager d'autres solutions que nous proposerons. Dans la syntaxe, il faut être rigoureux. Chaque **fonction Si** doit posséder deux points-virgules (;), pour les deux possibilités. Chaque parenthèse ouverte doit être fermée. Comme la première encadre les suivantes, elles doivent toutes être fermées à la fin.

Dans le cas précédent traité dans l'autre support, nous souhaitons justement envisager une autre possibilité. Lorsque la moyenne générale de l'élève est comprise entre 9,5 et 10, il doit aller en rattrapage. Une seule **fonction Si** ne suffit pas. Le raisonnement littéral est le suivant :

Si la moyenne est supérieure ou égale à 10, l'élève passe. Sinon, si sa moyenne est comprise entre 9,5 et 10, il part en rattrapage, sinon il redouble.

Le dernier Sinon, celui de la seconde **fonction Si**, sous-entend que les deux premières conditions n'ont pas été satisfaites. Si la moyenne n'est ni supérieure à 10, ni supérieure à 9,5, c'est qu'elle est nécessairement inférieure. Par déduction logique, du fait de l'élimination en cascade, nous en déduisons que l'élève redouble.

L'**opérateur compris entre** n'existe pas dans la syntaxe des **formules Excel**. Il s'exprime par l'enchaînement des critères. Si le premier (≥ 10) n'est pas vérifié et que dans le même temps, le second ($\geq 9,5$) est rempli, nous en déduisons que la note est comprise entre les deux.

L'**assistant fonction** n'est pas d'une aide précieuse pour une telle construction. C'est pourquoi nous proposons de bâtir la formule manuellement. C'est d'ailleurs la meilleure solution en général pour bien maîtriser les fonctions et leur syntaxe.

- Sélectionner la première cellule du précédent calcul, soit la **cellule E5**,
- Taper le **symbole =** pour débiter le calcul,



- Saisir le nom de la fonction conditionnelle suivi d'une parenthèse, soit **Si**(,
- Construire le premier critère, soit : **D5>=10**,
- Taper un point-virgule (;) pour passer dans la *branche Alors* de la première fonction Si,
- Entre guillemets, saisir le texte Admis, soit : **"Admis"**,
- Taper un point-virgule (;) pour passer dans la *branche Sinon* de la première fonction Si,
- Saisir de nouveau le nom de la fonction conditionnelle suivi d'une parenthèse, soit **Si**(,
- Construire le critère suivant : **D5>=9,5**,
- Taper un point-virgule (;) pour passer dans la *branche Alors* de la seconde fonction Si,
- Entre guillemets, saisir le texte Rattrapage, soit : **"Rattrapage"**,
- Taper un point-virgule (;) pour passer dans la *branche Sinon* de la seconde fonction Si,
- Saisir le texte Recalé entre guillemets, soit : **"Recalé"**,
- Fermer deux parenthèses pour les deux fonctions Si,
- Valider la formule par le raccourci clavier CTRL + Entrée,
- Puis, double cliquer sur la poignée de la cellule pour répliquer la logique sur la hauteur du tableau,

	C	D	E
	Elèves	Moyenne générale	Avis du conseil
	Marc	12	Admis
	Sophie	8	Recalé
	Fred	8	Recalé
	Anne	9	Recalé
	Christine	14	Admis
	Luc	12	Admis
	Roger	9,8	Rattrapage
	Tifanie	13	Admis
	Paul	9	Recalé

Comme l'illustre la capture ci-dessus, les résultats sont quasiment identiques aux précédents. Mais un nouveau cas est considéré. L'élève ayant obtenu une moyenne proche de 10 (9,8), se voit en sursis grâce au rattrapage. L'analyse du problème est devenue plus fine grâce à l'**imbrication des fonctions Si**, qui a permis d'augmenter le nombre de conditions à vérifier.

Et, Ou : comment recouper les conditions

Parfois, il est nécessaire de vérifier ensemble plusieurs conditions pour satisfaire un critère. Il ne s'agit plus de les déduire en cascade mais de les considérer en même temps.

- Cliquer sur l'**onglet Recouper_criteres** en bas de la fenêtre Excel pour activer sa feuille,

Ets l'Indulgent

	Math	12
Ou	Français	10

Français	12
Anglais	15
Mathématiques	8

Moyenne

Délégation
Terminale S

Ets Le Stricte

	Math	12
Et	Français	10

Français	12
Anglais	15
Mathématiques	8

Moyenne

Délibération
Terminale S

Une petite simulation est proposée sur cette feuille. Deux établissements scolaires sont mis en opposition. Ils n'ont pas les mêmes règles de fonctionnement et imposent des conditions de passage différentes. Pour passer en Terminale S avec l'établissement l'indulgent, il suffit d'avoir soit plus de 12 en Math, soit plus de 10 en Français. Avec l'établissement Le Stricte, l'élève doit avoir à la fois plus de 12 en Math et plus de 10 en Français.

Dans le premier cas, il suffit que l'un des critères soit vérifié pour que la condition globale de passage soit remplie. C'est la **fonction Excel Ou** qui permet d'énumérer les cas, autant que souhaité.

=Ou(Critère1 ; Critère2 ; ... ; CritèreN)

Dans le second cas, il faut que tous les critères soient vérifiés ensemble pour que la condition générale de passage soit satisfaite. Si l'un d'entre eux n'est pas vrai, la condition globale n'est pas remplie. C'est la **fonction Excel Et** qui permet d'énumérer ces cas, autant que souhaité.

=Et(Critère1 ; Critère2 ; ... ; CritèreN)

Ces fonctions permettent donc d'énumérer des critères. Pour qu'ils soient évalués, elles doivent être imbriquées dans une fonction Si.

=Si(Et(Critère1 ; Critère2 ; ... ; CritèreN) ; Alors Action1 ; Sinon Action2)

=Si(Ou(Critère1 ; Critère2 ; ... ; CritèreN) ; Alors Action1 ; Sinon Action2)

Commençons par le premier établissement.

- Sélectionner la **cellule D13** et taper le **symbole** = pour débiter le calcul,
- Saisir le nom de la fonction conditionnelle suivi d'une parenthèse, soit **Si**,
- Saisir la fonction signifiant *Soit*, suivie d'une parenthèse, soit : **Ou**,
- Cliquer sur la note obtenue en Math, soit la **cellule D9**,
- Taper le symbole supérieur suivi du symbole égal, soit : **>=**, pour initier la comparaison,
- Cliquer la cellule de la condition correspondante, soit la **cellule D4**,
- Taper un point-virgule (;) pour poursuivre l'énumération des critères,
- Cliquer sur la note obtenue en Français, soit sur la **cellule D7**,
- Taper le symbole supérieur suivi du symbole égal, soit : **>=**, pour initier la comparaison,
- Cliquer la cellule de la condition correspondante, soit la **cellule D5**,



- Fermer la parenthèse de la fonction Ou,
- Taper un point-virgule (;) pour passer dans la *branche Alors* de la fonction Si,
- Saisir le texte : Admis entre guillemets, soit : "**Admis**",
- Taper un point-virgule (;) pour passer dans la *branche Sinon* de la fonction Si,
- Saisir le texte : Recalé entre guillemets, soit : "**Recalé**",
- Fermer la parenthèse de la fonction Si et valider la formule,

Comme vous le remarquez, l'élève est admis. L'un des deux critères est vérifié. Seule la note en Français est supérieure au minimum requis. Et dans ce cas, même si cela ne semble pas très logique, la **fonction Si** considère que la condition générale est satisfaite. Alors, elle enclenche la première action prévue. L'issue est la même avec 8 en Français et 12 en Mathématiques ou avec 13 en Math et 12 en Français. Lorsque les deux critères sont vérifiés, la condition est bien entendu remplie aussi. Il faut qu'aucun des critères ne soit rempli pour que la **fonction Si** évalue la condition comme non satisfaite. Ainsi, lorsque l'élève obtient 9 en Français et 9 en Mathématiques, la **fonction Si** enclenche la deuxième action. L'élève est donc recalé.

La capture ci-dessous illustre le résultat.

B	C	D	E	F	G
	Math	12			Math
Ou	Français	10		Et	Français
	Français	12			Français
	Anglais	15			Anglais
	Mathématiques	8			Mathématiques
	Moyenne				Moyenne
	Délégation Terminale S	Admis			Délégation Terminale S

Nous devons désormais appliquer cette formule au deuxième cas, pour l'établissement Le Stricte. Les deux conditions doivent être vérifiées ensemble et non l'une après l'autre. La **fonction Ou** doit être remplacée par la **fonction exclusive Et**.

- En **cellule H13**, adapter le calcul comme suit :

=SI(ET(D9>=D4;D7>=D5);"Admis";"Recalé")

Pour les mêmes notes que précédemment, la décision est opposée, l'élève est recalé. L'un des deux critères n'est pas vérifié dans l'énumération de la **fonction Et**. La **fonction Si** considère la condition générale comme non satisfaite et enclenche la seconde action, celle de la *branche Sinon*.

Remarque, reprendre la saisie d'un calcul : Plusieurs techniques permettent de reprendre la saisie d'un calcul déjà validé. Vous pouvez par exemple modifier le calcul depuis la **barre de formule** en ayant



préalablement sélectionné la cellule. Vous pouvez aussi double cliquer sur la cellule. La formule s'affiche et vous pouvez ainsi la modifier directement sur la feuille. La **touche F2 du clavier** permet elle aussi de rendre active la saisie d'une cellule qui contient déjà des informations.

La concaténation

La **concaténation** est une opération qui consiste à assembler les chaînes de caractères, en les mettant bout à bout. Elle permet de joindre des textes, mais aussi et pourquoi pas, un texte à un résultat de calcul pour rendre sa valeur plus explicite.

L'**opérateur de concaténation** dans Excel est le **Et commercial** ou esperluette. Il s'obtient par la touche 1 du clavier, soit **&**. Il s'utilise dans une formule usuelle qui débute donc par le symbole égal (=).

- Cliquer sur l'**onglet Concaténer** en bas de la fenêtre Excel pour activer sa feuille,

Le tableau propose une petite base de données de clients. Ils sont décrits sur trois colonnes par leur civilité, leur nom et leur prénom. Vous l'avez compris, l'objectif est d'assembler ces trois informations dans la **colonne Assemblage**, dans l'ordre Civilité Prénom et Nom.

La concaténation - Assembler les chaînes de caractères			
Civilité	Nom	Prénom	Assemblage
Mr	Hochon	Paul	
Mme	Audine	Anne	
Mme	Rouana	Marie	
Mme	Hamalibou	Arlette	
Mr	Céhef	Hassan	
Mme	Galls	Charline	
Mr	Houda	Barrack	

Trois éléments sont à joindre. Nous devons donc utiliser deux caractères de **concaténation**. Le **Et commercial** intervient à chaque fois qu'il y a deux éléments à assembler.

- Sélectionner la première cellule de l'assemblage à réaliser, soit la **cellule E5**,
- Taper le symbole égal (=) pour débiter le calcul,
- Cliquer sur la civilité, soit la **cellule B5**,
- Taper le caractère de concaténation, soit : **&**,
- Cliquer sur le prénom, soit la **cellule D5**,
- Taper le caractère de concaténation, soit : **&**,
- Cliquer sur le nom, soit la **cellule C5**,
- Valider la formule par **Ctrl + Entrée**,
- Puis, double cliquer sur la poignée de la cellule pour la répliquer sur tous les clients,

L'assemblage est effectivement réalisé, mais un défaut saute aux yeux. Tous les termes sont parfaitement associés, dans l'ordre demandé, tel que nous l'avons écrit dans la formule. Mais ils ne sont pas séparés d'un espace. Un espace est un élément comme un autre. A chaque fois qu'il est ajouté pour séparer deux mots, il doit être associé par le caractère de concaténation. L'espace est un texte, il doit donc être introduit entre guillemets.

- En **E5**, modifier la formule comme suit :



=B5&" "&D5&" "&C5

- Valider le calcul par **Ctrl + Entrée**,
- Puis le répliquer par double clic sur la poignée de la cellule,

Police

Alignement

✕

✓

fx

=B5&" "&D5&" "&C5

B	C	D	E
Civilité	Nom	Prénom	Assemblage
Mr	Hochon	Paul	Mr Paul Hochon
Mme	Audine	Anne	Mme Anne Audine
Mme	Rouana	Marie	Mme Marie Rouana
Mme	Hamalibou	Arlette	Mme Arlette Hamalibou
Mr	Céhef	Hassan	Mr Hassan Céhef

Le résultat est plus net. Chaque terme est séparé d'un autre par un espace.

Dans la formule, entre chaque cellule, nous avons concaténé le texte avec un espace entre guillemets. Celui-ci est donc encadré par deux caractères de concaténation. Le premier sert à assembler le mot précédent avec l'espace qui suit. Le second permet de concaténer ce résultat avec le mot suivant.

Concaténation par fonction de calcul

Lorsque les termes à assembler sont nombreux, la répétition de la **concaténation** avec l'espace devient fastidieuse et alourdit la syntaxe. C'est pourquoi **Excel** propose une fonction permettant de définir une bonne fois pour toutes, le caractère de séparation. Cette fonction se nomme **Joindre.Texte**. Sa syntaxe est la suivante :

=Joindre.Texte(Séparateur ; Ignorer_si_vide ; Terme1 ; Terme2 ; ... ; TermeN)

Le deuxième argument est un **booléen**. Défini sur la valeur **Vrai**, il permet à la fonction d'ignorer les cellules vides pour ne pas dupliquer les séparateurs comme l'espace.

- En **cellule E5**, taper le **symbole =** pour réinitialiser le calcul,
- Saisir le nom de la fonction de concaténation suivi d'un espace, soit **Joindre.Texte(**,
- Saisir un espace entre guillemets pour définir le séparateur, soit : " " ,
- Taper un point-virgule (;) pour passer dans l'argument suivant de la fonction,
- Saisir le texte **Vrai** pour lui indiquer d'ignorer les potentielles cellules vides,
- Taper un point-virgule (;) pour passer dans l'argument du premier texte à assembler,
- Cliquer sur la civilité, soit la **cellule B5**,
- Taper un point-virgule (;) pour passer dans l'argument du deuxième texte à assembler,
- Cliquer sur le prénom, soit la **cellule D5**,
- Taper un point-virgule (;) pour passer dans l'argument du dernier texte à assembler,
- Cliquer sur le nom, soit la **cellule C5**,
- Fermer la parenthèse de la fonction Joindre.Texte,
- Valider le calcul par le raccourci Ctrl + Entrée,
- Double cliquer sur la poignée de la cellule pour le répliquer sur tout le tableau,



Police				Alignement
=JOINDRE.TEXTE(" ";VRAI;B5;D5;C5)				
B	C	D	E	
Civilité	Nom	Prénom	Assemblage	
Mr	Hochon	Paul	Mr Paul Hochon	
Mme	Audine	Anne	Mme Anne Audine	
Mme	Rouana	Marie	Mme Marie Rouana	
Mme	Hamalibou	Arlette	Mme Arlette Hamalibou	
Mr	Céhef	Hassan	Mr Hassan Céhef	
Mme	Galle	Charline	Mme Charline Galle	

Le résultat est strictement identique, comme l'illustre la capture ci-dessus. Mais la formule de concaténation est plus simple. Et cette remarque est d'autant plus vraie que le nombre de termes à associer augmente.

Nous n'avons spécifié qu'une seule fois le caractère de séparation, soit l'espace entre guillemets, en premier paramètre. Après l'argument booléen, nous nous sommes contentés d'énumérer les termes à associer, séparés d'un point-virgule, selon la syntaxe Excel.

Résultat statistique et concaténation

La **fonction Excel Nb.Si** est particulièrement intéressante car elle permet de compter, sur une plage de cellules donnée, le nombre de fois où le critère défini est vérifié. Elle requiert donc deux paramètres : La plage de cellules sur laquelle le critère doit être dénombré et le critère lui-même.

=Nb.Si(Plage_de_cellules ; Critère_à_compter)

Elle fournit un résultat numérique brut qu'il est souvent judicieux d'explicitier par concaténation.

- Cliquer sur l'**onglet Conditions** en bas de la fenêtre Excel pour revenir sur sa feuille,

Nous proposons de remplacer le résultat de la moyenne, précédemment calculé en cellule E15, par le résultat statistique du nombre d'admis.

- Cliquer sur la **cellule E15** pour l'activer,
- Taper le **symbole =** pour débiter le calcul,
- Saisir le nom de la fonction de dénombrement suivi d'une parenthèse, soit : **Nb.Si(**,
- Sélectionner toutes les cellules de la colonne Avis du conseil, soit la plage : **E5:E13**,
- Taper un point-virgule (;) pour passer dans l'argument du critère,
- Cliquer sur la **cellule H3** du panneau de contrôle,
- Fermer la parenthèse de la fonction Nb.Si,
- Puis, valider le calcul,



fx **=NB.SI(E5:E13;H3)**

Conseil de classe		
Elèves	Moyenne générale	Avis du conseil
Marc	12	Admis
Sophie	8	Recalé
Fred	8	Recalé
Anne	9	Recalé
Christine	14	Admis
Luc	12	Admis
Roger	9,8	Rattrapage
Tifanie	13	Admis
Paul	9	Recalé

Panneau de contrôle	
10	Admis
>=	

5	4
---	---

Synthèse
4

Le résultat tombe et il est cohérent. Cette valeur est dynamique puisqu'elle se mettra à jour si les données du tableau venaient à évoluer. Elle indique que 4 élèves sont admis sur l'ensemble de la classe. Cette donnée statistique est évidente pour le concepteur. Mais elle peut nécessiter réflexion pour toute autre personne. Pour éviter toute ambiguïté, nous proposons donc de commenter ce résultat par concaténation.

- Double cliquer sur la **cellule E15** pour activer la saisie du calcul,
- Cliquer à la toute fin de la formule après la parenthèse fermante de la fonction Nb.Si,
- Taper le symbole de concaténation, soit : **&**,
- Saisir un espace entre guillemets, soit : **" "**,
- Taper de nouveau le symbole de concaténation, soit : **&**,
- Puis, cliquer sur la cellule H3 du panneau de contrôle,
- Valider la modification en enfonçant la touche Entrée du clavier,

La formule que nous avons construite est la suivante :

=NB.SI(E5:E13;H3)&" "&H3

fx **=NB.SI(E5:E13;H3)&" "&H3**

Conseil de classe		
Elèves	Moyenne générale	Avis du conseil
Marc	12	Admis
Sophie	8	Recalé
Fred	8	Recalé
Anne	9	Recalé
Christine	14	Admis
Luc	12	Admis
Roger	9,8	Rattrapage
Tifanie	13	Admis
Paul	9	Recalé

Panneau de contrôle	
10	Admis
>=	

5	4
---	---

Synthèse
4 Admis



Nous récupérons le calcul précédent. Nous le concaténons tout d'abord avec un espace (&" "), pour séparer les informations. Puis, nous concaténons cet espace avec un texte explicatif (&H3). Nous obtenons ainsi le résultat : **4 Admis**. L'information devient évidente et pertinente.

La recherche d'informations

Excel propose entre autres, deux fonctions très puissantes pour [extraire de l'information de bases de données selon critère](#). Il s'agit des fonctions **RechercheV** et **RechercheH**. La première effectue une recherche verticale tandis que la seconde effectue une recherche horizontale. Chacune est adaptée à une disposition différente des données. La plus courante est la **fonction RechercheV**. Les tableaux sont le plus souvent organisés sous forme de colonnes. Dans certains cas, elles peuvent remplacer les **fonctions Si**, lorsque le problème propose beaucoup de conditions à analyser.

- Cliquer sur l'**onglet Chercher-Extraire** en bas de la fenêtre Excel pour activer sa feuille,

Le tableau énumère une liste de véhicules d'un parc automobile. Chaque voiture est détaillée sur trois colonnes (Immatriculation, Marque et Modèle).

Comme cette base de données peut être volumineuse, l'objectif consiste à simplifier la recherche d'un véhicule sur son immatriculation. L'utilisateur tape une immatriculation. A validation, toutes les références du véhicule sont rapatriées dans des cellules prévues à cet effet. Le gain de temps est conséquent.

Comme l'illustre la capture ci-dessous, nous avons prévu une [liste déroulante en cellule F5](#), pour simplifier la sélection d'une référence.

biles - Base de données		
	Modèle	
	MODUS	
	TWINGO (07/1998-09/2000)	
N	GOLF (01/1998-03/2004)	
	208 1.2 PURETECH ACTIVE 5P	
	308 ACTIVE 1.6 HDI 92CV 5P	
	308 ACTIVE 1.6 HDI 92CV 5P	

Choix Immat.
112CSB59
1702WK74
170XF74
2020JNY
2043JNY
2069JNY
2416VK03
2677XT74

Juste en dessous, en cellules G8 et G9, nous avons prévu une zone d'extraction pour réceptionner les informations correspondant à la demande effectuée depuis cette **liste déroulante**.

Pour récupérer et afficher la marque et le modèle du véhicule, nous devons réaliser une recherche de l'immatriculation dans le tableau de données. Il s'agit d'exploiter la **fonction Excel RechercheV**, selon la syntaxe suivante :

=RechercheV(Valeur_cherchée ; Tableau_de_recherche ; Colonne_de_retour ; Faux)

La valeur cherchée est la cellule de l'immatriculation choisie, soit **F5**. Le tableau de recherche est celui du parc automobile, soit la **plage de cellules B4:D92**. La colonne de retour est le numéro de colonne dans laquelle se trouve l'information à récupérer. Par exemple, la marque se trouve en deuxième



colonne (2) du tableau de recherche. Le dernier argument est un **booléen** que nous fixerons à **Faux** pour réaliser une recherche selon une correspondance exacte.

- Sélectionner une immatriculation à l'aide de la liste déroulante, par exemple : 2416VK03,
- Cliquer sur la cellule de la marque à récupérer, soit la **cellule G8**,
- Taper le **symbole** = pour débiter le calcul,
- Saisir le nom de la fonction d'extraction suivi d'une parenthèse, soit **RechercheV**,
- Cliquer sur la **cellule F5** pour désigner l'élément cherché, soit l'immatriculation,
- Taper un point-virgule (;) pour passer dans l'argument du tableau de recherche,
- Sélectionner l'intégralité de la base de données, soit la **plage de cellules B4:D92**,
- Taper un point-virgule (;) pour passer dans l'argument du numéro de colonne,
- Saisir le **chiffre 2** pour désigner la colonne Marque,
- Taper un point-virgule (;) suivi de la valeur **Faux** pour une recherche exacte,
- Fermer la parenthèse de la fonction RechercheV,
- Enfin, valider le calcul à l'aide de la touche Entrée,

La marque du véhicule apparaît instantanément. Si vous choisissez une autre immatriculation avec la liste déroulante, la marque correspondante est aussitôt extraite et affichée. La **fonction RechercheV** est donc une fonction d'extraction très puissante à la syntaxe relativement simple. Elle impose néanmoins une limitation. L'élément cherché (L'immatriculation ici) doit nécessairement figurer en première colonne du tableau de recherche. Soit il s'agit de préparer le tableau en conséquence, soit il faut utiliser les [fonctions de recherche Index et Equiv](#), le cas échéant.

Pour extraire le modèle selon l'immatriculation, la formule est quasiment identique. Il s'agit toujours d'effectuer la recherche de l'immatriculation dans la base de données des véhicules. Seul le troisième argument change. La colonne de retour n'est plus la deuxième (2) mais la troisième, soit l'indice 3 par rapport à la sélection.

- En cellule G9 pour le modèle, adapter le calcul comme suit :

=RECHERCHEV(F5;B4:D92;3;FAUX)

- Puis, valider la formule,

Le modèle correspondant est aussitôt extrait. Comme précédemment, si vous choisissez une nouvelle immatriculation, les deux informations rapatriées se mettent instantanément à jour.

Un défaut subsiste néanmoins. Si vous supprimez l'élément de recherche en cellule F5, les deux **fonctions RechercheV** retournent un message d'erreur : **#N/A**. Il signifie littéralement : Not Available, soit non disponible. Les fonctions d'extraction réagissent ainsi en effet lorsque l'élément de recherche n'est pas trouvé. Pour palier le souci, nous proposons d'encadrer la recherche dans une fonction Excel de gestion d'erreur. Cette dernière se nomme **SiErreur** et sa syntaxe est la suivante :

=SiErreur(Calcul_à_essayer ; Gestion_d_erreur)

En premier argument, nous devons lui passer le calcul de la recherche. S'il fonctionne, le résultat est retourné. S'il génère une erreur, c'est le deuxième argument de la **fonction SiErreur** qui est appelé. Il suffit d'inscrire un espace entre guillemets pour conserver la cellule vide lorsqu'aucune immatriculation n'est choisie.

- En cellule G8 et G9, adapter les calculs comme suit :

=SIERREUR(RECHERCHEV(F5;B4:D92;2;FAUX);"")
=SIERREUR(RECHERCHEV(F5;B4:D92;3;FAUX);"")

Police) Alignement Nombre

✓ f1 =SIERREUR(RECHERCHEV(F5;B4:D92;2;FAUX);"")

	E	F	G
		Choix Immat.	
		8033JPB	
9/2000)			
004)		Résultats	
CTIVE 5P		Marque	PEUGEOT
2CV 5P		Modèle	308 ACTIVE 1.6 HDI 92CV 5P
2CV 5P			

Les deux messages d'erreur disparaissent aussitôt. Mais si vous désignez de nouveau une immatriculation, l'extraction se produit correctement. Le résultat est plus propre.

Remarque, comment afficher les formules dans les cellules à la place des résultats ?

Il faut dans un premier temps cliquer sur l'**onglet Formules** en haut de la fenêtre Excel pour activer son ruban. Il suffit ensuite de cliquer sur le **bouton Afficher les formules** dans la section *Vérification des formules* du ruban. Les colonnes se redimensionnent pour accueillir la syntaxe complète des calculs. Mais si vous cliquez de nouveau sur le bouton, les résultats réapparaissent et la structure de la feuille se rétablit.

Références relatives - références absolues

Nous abordons désormais une notion fondamentale d'**Excel**. Jusqu'alors, nous avons exclusivement raisonné en **références relatives**. Cela signifie que les coordonnées des cellules sont adaptées en même temps que le calcul est répliqué, pour ajuster la logique sur les nouvelles cellules. Dans l'exemple de la **feuille Conditions**, le critère de la fonction Si est posé sur la **cellule D5**. Il s'agit de la première moyenne générale en regard de la cellule du calcul. Et lorsque nous avons reproduit la formule sur les lignes du dessous avec la poignée, **Excel** a adapté les références du critère, pour que chaque moyenne puisse être évaluée. La condition est ainsi vérifiée sur la cellule D6 puis D7 et D8 etc... Les références suivent le sens du déplacement.

Mais certains calculs doivent intégrer des cellules de repère, qui ne doivent pas bouger. **Excel** permet de figer une cellule dans un calcul grâce à la touche F4 du clavier. Les coordonnées de la cellule ainsi fixée ne changent plus, bien que le calcul soit répliqué. C'est ce que nous proposons de démontrer ici.

- Cliquer sur l'**onglet Références absolues** en bas de la fenêtre Excel pour activer sa feuille,

Ce tableau énumère les quantités achetées pour certains produits. Chacun d’entre eux est associé à un prix unitaire (Au kg) hors taxes en colonne D. Il s’agit dans un premier temps de calculer le montant total des ventes par produit, en colonne F. Le résultat conduira donc à un total hors taxes. Puis, dans un deuxième temps, l’objectif consiste à calculer un total Ttc par produit, en colonne I. Pour cela, la TVA inscrite en cellule I2 de référence, doit être exploitée dans le calcul.

- Sélectionner la **cellule F5** et taper le **symbole =** pour débiter le calcul,
- Cliquer le premier prix unitaire, soit la **cellule D5**,
- Taper le symbole de l'étoile (*) pour la multiplication,
- Cliquer la première quantité, soit la **cellule E5**.



- Valider le calcul par le raccourci clavier CTRL + Entrée,
- Puis, tirer la poignée du résultat vers le bas jusqu'en **cellule F9**,

=D9*E9			
C	D	E	F

Produits à la vente			
ation	PUHT	Qté	Total
es	4,50 €	18	81,00 €
	6,12 €	5	30,60 €
s	7,50 €	32	240,00 €
s	3,12 €	16	49,92 €
es	6,90 €	15	=D9*E9
Total			

Tous les montants hors taxes sont instantanément reproduits dans la colonne. Sans le savoir vous venez d'exploiter les références relatives. Il s'agit du fonctionnement intuitif d'Excel par défaut. Nous avons posé un calcul sur la ligne 5 (en F5) pour multiplier deux cellules de cette même ligne. Lorsque nous avons validé puis reproduit la logique avec la poignée, nous avons demandé à Excel de répliquer le calcul sur les lignes du dessous (6, 7, 8 et 9).

Sur la ligne 9 illustrée par la capture ci-dessus, **Excel** ne multiplie plus les cellules de la ligne 5 mais bien les cellules de la ligne en cours (D9*E9). C'est pourquoi on dit que les références du calcul sont relatives. Si le calcul bouge, en changeant de ligne comme ici, les références suivent le déplacement. Cette remarque est évidemment vraie pour un déplacement en colonne.

Dans la majorité des cas, les références relatives sont adaptées au contexte du calcul comme dans cet exemple. Mais dans d'autres circonstances, comme nous allons le voir avec le calcul du TTC, ce comportement n'est plus adapté.

Le résultat du TTC consiste à calculer le pourcentage dû à la TVA sur le montant total hors taxes (THT*TVA), puis à l'ajouter au total hors taxes de départ. Il peut se traduire sous les deux formes suivantes :

$$=THT*TVA + THT$$

$$=THT*(TVA + 1)$$

Les deux sont identiques. Dans le deuxième cas, on parle de factorisation. Nous proposons d'exploiter la première forme pour plus de clarté.

- Sélectionner la cellule du premier TTC, soit la **cellule I5**,
- Taper le **symbole =** pour débiter le calcul,



- Cliquer le premier total hors taxes, soit la **cellule F5**,
- Taper le symbole de l'étoile (*) pour la multiplication,
- Cliquer sur la TVA, soit sur la **cellule I2**,
- Taper le **symbole +** pour l'addition,
- Cliquer de nouveau sur le premier total hors taxes, soit la **cellule F5**,
- Valider le calcul par le raccourci CTRL + Entrée,

Le premier résultat tombe et il est cohérent. Le montant est supérieur au total hors taxes de départ. Il a été majoré de 20%.

- Tirer la poignée du calcul jusqu'en **cellule I9**,

E		F	G	H	I
Vente		TVA		20%	
	Qté	Total	Montant taxe	Total TTC	
	18	81,00 €		97,20 €	
	5	30,60 €		30,60 €	
	32	240,00 €		#VALEUR!	
	16	49,92 €		4 902,14 €	
	15	103,50 €		=F9*I6+F9	
Total					

Cette fois, rien ne va plus. Certains TTC sont identiques au total hors taxes de départ. D'autres sont exorbitants et d'autres encore conduisent à un message d'erreur (#VALEUR!). La raison est due à ces fameuses **références relatives**, qui en l'absence d'indication contraire, ont adapté les références des cellules au déplacement du calcul.

Comme l'illustre la capture ci-dessus, pour le calcul du dernier Ttc, c'est bien le dernier total hors taxes situé sur la même ligne, qui a été pris en compte. Il s'agit de F9 et non plus de F5. Jusque-là, le déplacement est logique et souhaité. Mais la Tva est unique. C'est pourquoi elle est inscrite dans une cellule de référence, en I2. Or, lorsque nous répliquons la formule, **Excel** adapte ses références en fonction du déplacement. I2 devient I3 puis I4 jusqu'à I6. De fait, nous obtenons des résultats conséquents en multipliant un Tht par un Ttc ou une erreur, en multipliant un Tht par le titre Total TTC. Nous devons indiquer à **Excel** que cette TVA est une référence qui ne doit pas bouger. Nous l'avons dit plus haut, c'est la **touche F4 du clavier** qui permet de fixer une cellule dans un calcul. On parle de **références absolues**.

- Sélectionner de nouveau la première cellule pour le calcul du TTC, soit **I5**,
- Taper le **symbole =** pour démarrer la formule,
- Cliquer sur le premier total hors taxes, soit sur la **cellule F5**,
- Taper le symbole de l'étoile (*) pour la multiplication,
- Cliquer sur la TVA, soit sur la **cellule I2**,
- Enfoncer la **touche F4 du clavier**,

Vous remarquez l'apparition de deux dollars (\$) venant encadrer les références de la cellule (\$I\$2). Elle est désormais figée en ligne et en colonne. Cette **référence absolue** ne bougera plus.

- Taper le **symbole +** pour l'addition,



- Puis cliquer de nouveau sur le premier total hors taxes, soit sur la **cellule F5**,
- Valider le calcul par le raccourci CTRL + Entrée,
- Puis, tirer la poignée du résultat jusqu'en **cellule I9**,

=F9*\$I\$2+F9

Produits à la vente				TVA	
					20%
ation	PUHT	Qté	Total	Montant taxe	Total TTC
es	4,50 €	18	81,00 €		97,20 €
	6,12 €	5	30,60 €		36,72 €
	7,50 €	32	240,00 €		288,00 €
	3,12 €	16	49,92 €		59,90 €
es	6,90 €	15	103,50 €		=F9*\$I\$2+F9
			Total		

Le problème est résolu. Comme l'illustre la capture ci-dessus, le dernier calcul en **I9** considère bien le total hors taxes correspondant situé en **F9**. En revanche, il continue de considérer le même taux de TVA que les autres, situé en **I2**. Les références n'ont pas suivi le sens du déplacement grâce aux dollars ajoutés par la touche F4. Cette notion est fondamentale pour pouvoir bâtir des calculs sur des variables, soit des valeurs inscrites dans des cellules. Si le taux de Tva venait à changer, il suffirait de le modifier en cellule I2. En le passant à 22% par exemple, vous constatez que tous les résultats se mettent à jour. Les **calculs sont dynamiques dans Excel**. Forts de ces acquis, nous corrigerons bientôt les précédents calculs statiques que nous avons construits. C'était le cas pour la délibération du conseil selon la moyenne générale, dans la **feuille Conditions**. Le critère était bâti sur une constante (10) au lieu d'une cellule variable. De même, les actions consistaient en des textes inscrits entre guillemets alors que nous devons considérer des cellules là encore.

Remarque, deux dollars (\$) pour deux degrés de libertés : Lorsque vous enfoncez la **touche F4 du clavier**, deux dollars encadrent les références de la cellule. Un dollar se place devant l'indice de colonne (\$I\$2). Il bloque le déplacement vers la droite et la gauche. Un autre se place devant l'indice de ligne (\$I\$2). Il bloque le déplacement vers le haut et le bas. Si vous enfoncez de nouveau la touche F4 (en saisie de formule), vous constatez qu'il ne subsiste plus qu'un dollar devant la référence de ligne (I\$2). Si vous continuez d'enfoncer la touche F4, le dollar se positionne devant la référence de colonne (\$I2). Et si vous enfoncez une dernière fois la touche F4, les dollars disparaissent rendant les deux degrés de liberté à la cellule (ligne et colonne). Dans certains calculs particuliers en effet, il est nécessaire de déplacer la cellule dans une direction, mais pas dans l'autre.

Les calculs conditionnels

Excel propose des fonctions puissantes permettant de réaliser des opérations selon des conditions à vérifier. Ainsi, nous avons déjà abordé la **fonction Nb.Si** capable de compter toutes les cellules d'une plage répondant à un critère spécifique. Mais il est aussi possible de réaliser des moyennes ou des sommes n'intégrant dans le calcul que les cellules satisfaisant la condition. La **fonction Somme.Si** permet par exemple de réaliser des **sommes conditionnelles**. La [formation Excel pour réaliser des classements dynamiques la met en valeur](#). Sa syntaxe est la suivante :

=Somme.Si(Plage_du_critère ; Critère ; [Plage_pour_somme])



Elle requiert trois arguments, le dernier étant facultatif. Il s'agit tout d'abord de désigner la plage de cellules sur laquelle doit être cherché le critère. Il s'agit ensuite de spécifier ce critère. Enfin, il faut indiquer la plage de cellules correspondante sur laquelle l'addition doit être réalisée. Lorsqu'un critère est vérifié sur la première plage, la valeur correspondante sur la seconde est additionnée. Si cette dernière plage n'est pas renseignée, la **fonction Somme.Si** utilise la première plage pour le critère et la somme.

- Cliquer sur l'**onglet Calculs_conditionnels** en bas de la fenêtre Excel pour activer sa feuille,

Le tableau énumère les tarifs de certains matériels multimédias. Ils appartiennent tous à une catégorie référencée en colonne B (Cat.). Dans le petit tableau de synthèse situé entre les lignes 14 et 16, nous souhaitons afficher la somme des tarifs par catégorie. Cette somme doit être dynamique. Il s'agit donc de chercher le critère sur la plage de cellules de la colonne B et de réaliser la somme sur celle de la colonne D.

- Sélectionner la première cellule du calcul de synthèse, soit la **cellule D14**,
- Taper le **symbole =** pour débiter la formule,
- Saisir la fonction de l'addition conditionnelle suivie d'une parenthèse, soit **Somme.Si(**,
- Sélectionner la **plage de cellules B5:B12** pour désigner la zone du critère,
- Enfoncer la **touche F4 du clavier** pour figer ses bornes, ce qui donne : **\$B\$5:\$B\$12**,
- Taper un point-virgule (;) pour passer dans l'argument du critère,
- Désigner la catégorie à chercher en spécifiant la **cellule C14**,

Si vous ne pouvez pas la cliquer, vous pouvez saisir ses références.

- Taper un point-virgule (;) pour passer dans l'argument de la plage pour l'addition,
- Sélectionner les montants hors taxes soit la **plage de cellules D5:D12**,
- Enfoncer la touche F4 du clavier pour figer ses bornes, ce qui donne : **\$D\$5:\$D\$12**,
- Fermer la parenthèse de la fonction Somme.Si,
- Valider le calcul avec le raccourci clavier CTRL + Entrée,
- Puis, tirer la poignée du calcul sur les deux cellules du dessous,

Police		Alignement			
		=SOMME.SI(\$B\$5:\$B\$12;C16;\$D\$5:\$D\$12)			
B	C	D	E	F	
Cat.	Des.	HT			
Cat1	Ecran LCD	890,00 €			
Cat1	Pc Ultron	950,00 €			
Cat2	Livre Junior	15,00 €			
Cat3	Dvd Pack 4	21,00 €			
Cat1	Impr. Desk4	120,00 €			
Cat2	BD Jack M	12,00 €			
Cat3	BluRay RM	19,00 €			
Cat2	Pochettes Pst	6,00 €			

Cat1	1 960,00 €
Cat2	33,00 €
Cat3	=SOMME.SI(\$B\$5:\$B\$12;C16;\$D\$5:\$D\$12)

SOMME.SI(plage; critère; [somme_plage])



Nous obtenons tous les montants par catégorie. Aucune constante n'apparaît dans ce calcul. Les résultats sont donc dynamiques. Si vous interchangez certaines catégories en colonne B, vous remarquez la mise à jour instantanée des sommes conditionnelles, dans le petit tableau de synthèse. La formule que nous avons construite est la suivante :

`=SOMME.SI(B5:B12;C14;D5:D12)`

Grâce aux **références absolues** que nous venons d'apprendre, nous avons pu bâtir une seule formule capable de se répliquer sur les autres cellules. On ne refait jamais deux fois le même calcul dans **Excel**. Si nous n'avions pas figé les plages de cellules, leurs bornes se seraient déplacées pour suivre le mouvement du calcul. Les résultats auraient été erronés puisque des catégories auraient été omises. Le critère (C14) en revanche ne devait justement pas être figé. Il change naturellement la catégorie à trouver en même temps que le calcul se déplace sur les lignes du dessous.

Conditions et références absolues

Maintenant que nous maîtrisons les techniques permettant de réaliser des **calculs dynamiques**, nous proposons d'améliorer la formule que nous avons bâtie pour statuer selon les moyennes générales.

- Cliquer sur l'**onglet Conditions** en bas de la fenêtre Excel pour revenir sur sa feuille,

Pour que ce modèle puisse être appliqué par différents établissements, nous devons remplacer les constantes du calcul par des variables. La formule que nous avons créée est la suivante :

`=SI(D5>=10;"Admis";SI(D5>=9,5;"Rattrapage";"Recalé"))`

Les établissements scolaires n'ont pas tous les mêmes conditions de passage. La **constante 10** dans le critère, doit être remplacée par la référence du panneau de contrôle en **G3**. Les termes employés peuvent différer eux aussi. Il s'agit de remplacer les expressions entre guillemets par leurs homologues, respectivement en cellules **H3**, **J3** et **K3**.

- Double cliquer sur la **cellule E5** pour activer la modification de la formule,
- Dans le premier critère, supprimer la constante 10,
- A la place, cliquer sur la **cellule G3** du panneau de contrôle,
- La figer avec la **touche F4 du clavier**,
- Dans la première *branche Alors*, supprimer le texte Admis et ses guillemets,
- A la place, cliquer sur la **cellule H3** et la figer,
- Supprimer la constante 9,5 dans le second critère,
- A la place, cliquer sur la **cellule I3** et la figer,
- Dans la *branche Alors* de la seconde fonction Si, supprimer le texte Rattrapage et ses guillemets,
- A la place, cliquer sur la **cellule J3** et la figer,
- Dans la *branche Sinon* de la seconde fonction Si, supprimer le texte Recalé et ses guillemets,
- A la place, cliquer sur la **cellule K3** et la figer,
- Valider le calcul par le raccourci CTRL + Entrée,
- Puis le répliquer sur la hauteur du tableau en double cliquant sur la poignée de sa cellule,

Les résultats obtenus sont strictement identiques. La différence est pourtant majeure. Ils sont désormais dynamiques.

- En **G3** saisir le nombre **13** pour modifier la condition de passage,
- En **H3**, remplacer la décision par le texte **Reçu**,



Alignement

Nombre

=SI(D5>=\$G\$3;\$H\$3;SI(D5>=\$I\$3;\$J\$3;\$K\$3))

C	D	E	F	G	H
Conseil de classe				Panneau de contrôle	
				13	Reçu

Elèves	Moyenne générale	Avis du conseil
Marc	=SI(D5>=\$G\$3;\$H\$3;SI(D5>=\$I\$3;\$J\$3;\$K\$3))	
Sophie	8	Recalé
Fred	8	Recalé
Anne	9	Recalé
Christine	14	Reçu
Luc	12	rattrapage
Roger	9,8	rattrapage
Tifanie	13	Reçu
Paul	9	Recalé

5	4	62
---	---	----

Synthèse

2 Reçu

Tous les résultats intègrent automatiquement ces nouvelles conditions pour se mettre à jour. Un élève n'est plus admis mais reçu. Désormais, il faut obtenir au moins 13 de moyenne générale pour réussir. Toutes les notes comprises entre 10 et 12 basculent en rattrapage. Et comme l'indique le petit résultat de synthèse en dessous du tableau (Nb.Si), seuls deux étudiants sont désormais reçus.

Dans ce dernier cas de figure, il serait même opportun d'exploiter la **concaténation** pour réaliser l'accord. Ce dernier doit être fait lorsque le résultat est supérieur à 1. Donc une **fonction Si** serait nécessaire pour vérifier la condition. Vous le constatez, toutes les notions acquises dans ce support, sont précieuses à plus d'un titre.